Maple: Práca s maticami

Maple poskytuje podporu pre prácu a operácie s maticami, na numerickej aj na symbolickej úrovni. Algoritmy špeciálnych knižníc, ako sú **NAG, LaPACK**, **ATLAS** a **MKL**, sa využívajú na zabezpečenie presných numerických výpočtov súvisiacich s maticovými operáciami. Pomocou palety príkazov pre vytvorenie matice môžeme pohodlne definovať ľubovoľnú maticu typu napr. aj 1000 x 1000 s prvkami v tvare desatinných čísel. Základným príkazom pre definíciu matice je príkaz

> Matrix(r, c, init, ro, sc, sh, st, o, dt, f, a)

Všetky parametre príkazu sú nepovinné. Systém však potrebuje získať dostatok informácií, aby mohol vytvoriť štruktúru matice. Ak nie sú uvedené žiadne špecifikácie, odpoveďou systému je matica typu **0 x 0**.

Príkaz

> Matrix(r)

vytvorí maticu typu r x r, ktorej prvky sú implicitne definované ako nuly.

> Matrix(2);

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Príkaz v tvare

> Matrix(r,c)

vytvorí maticu typu r x c, ktorej prvky sú implicitne definované ako nuly.

> Matrix(2,3);

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Niekoľko ďalších príkladov pre definíciu matíc:

> Matrix(1..3,1..2,5);

$$\begin{bmatrix} 5 & 5 \\ 5 & 5 \\ 5 & 5 \end{bmatrix}$$

> Matrix([[1,2,3],[4,5,6]]);

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

```
> f:= (i,j) -> x^(i+j-1):
> Matrix(2,f);
```

$$\begin{bmatrix} x & x^2 \\ x^2 & x^3 \end{bmatrix}$$

V poslednom uvedenom príklade sú prvky matice definované ako mocninové funkcie premennej *x*, pričom jej stupeň určuje umiestnenie prvku matice, čo je usporiadaná dvojica čísel (*i*, *j*), pre *i*, *j* = 1,2.

Maticu môžeme definovať pomocou priraďovacieho operátora. Maticu označíme ľubovoľnou premennou a potom sa na ňu odvolávame pod týmto názvom.

```
> A:=Matrix(3,4,[[1,2,3],[4,5,6]],readonly=true);
```

$$A := \begin{bmatrix} 1 & 2 & 3 & 0 \\ 4 & 5 & 6 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> with(LinearAlgebra):
V := <<1,2,3>|<4,5,6>|<7,8,9>|<10,11,12>>;
```

$$V := \begin{bmatrix} 1 & 4 & 7 & 10 \\ 2 & 5 & 8 & 11 \\ 3 & 6 & 9 & 12 \end{bmatrix}$$

```
> MA := Matrix([[9,9,9,9],[9,9,9,9],[9,9,9,9],[9,9,9,9]]);
```

$$MA := \begin{bmatrix} 9 & 9 & 9 & 9 \\ 9 & 9 & 9 & 9 \\ 9 & 9 & 9 & 9 \\ 9 & 9 & 9 & 9 \end{bmatrix}$$

Operácie s maticami

Súčin matice a skaláru – čísla

Použijeme vyššie definovanú maticu A a vynásobíme ju skalárom 2, resp. 3/2.

Operáciu zapíšeme jednoducho znakom určeným pre násobenie - *

```
> B:=2*A;
```

$$B := \begin{bmatrix} 2 & 4 & 6 & 0 \\ 8 & 10 & 12 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

> C:=(3/2)*A;

$$C := \begin{bmatrix} \frac{3}{2} & 3 & \frac{9}{2} & 0 \\ 6 & \frac{15}{2} & 9 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Operácia sa dá vykonať aj s formálnou premennou, napr. $\tilde{\lambda}$, *c*, ale použitím iného príkazu.

> with(LinearAlgebra):
F:=ScalarMultiply(A,lambda/c);

$$F := \begin{bmatrix} \frac{\lambda}{c} & \frac{2\lambda}{c} & \frac{3\lambda}{c} & 0 \\ \frac{4\lambda}{c} & \frac{5\lambda}{c} & \frac{6\lambda}{c} & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Súčet, súčin a rozdiel matíc

Definujeme matice, ktorých prvky sú formálne premenné a vypočítame postupne ich súčet, súčin a rozdiel.

> M1:=Matrix([[cos(k*Pi*v),sin(k*Pi*v),0,0],[-sin(k*Pi*v),cos(k*Pi*v),0,0],[0,0,1,0],[0,0,0,1]]);

> M2:=Matrix([[1,0,0,0],[0,cos(l*Pi*v),sin(l*Pi*v),0],[0,-sin(l*Pi*v),cos(l*Pi*v),0],[0,0,0,1]]);

$$M1 := \begin{bmatrix} \cos(k\,\pi\,v) & \sin(k\,\pi\,v) & 0 & 0 \\ -\sin(k\,\pi\,v) & \cos(k\,\pi\,v) & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$M2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \cos(l\,\pi\,v) & \sin(l\,\pi\,v) & 0 \\ 0 & -\sin(l\,\pi\,v) & \cos(l\,\pi\,v) & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

> M:=M1+M2;

$$M := \begin{bmatrix} \cos(k\,\pi\,v)+1 & \sin(k\,\pi\,v) & 0 & 0 \\ -\sin(k\,\pi\,v) & \cos(k\,\pi\,v)+\cos(l\,\pi\,v) & \sin(l\,\pi\,v) & 0 \\ 0 & -\sin(l\,\pi\,v) & 1+\cos(l\,\pi\,v) & 0 \\ 0 & 0 & 0 & 2 \end{bmatrix}$$

```
> N:=M1.M2;
```

$$N := \begin{bmatrix} \cos(k\,\pi\,v) & \sin(k\,\pi\,v)\cos(l\,\pi\,v) & \sin(k\,\pi\,v)\sin(l\,\pi\,v) & 0 \\ -\sin(k\,\pi\,v) & \cos(k\,\pi\,v)\cos(l\,\pi\,v) & \cos(k\,\pi\,v)\sin(l\,\pi\,v) & 0 \\ 0 & -\sin(l\,\pi\,v) & \cos(l\,\pi\,v) & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> K:=M1-M2;
```

$$K := \begin{bmatrix} \cos(k\,\pi\,v)-1 & \sin(k\,\pi\,v) & 0 & 0 \\ -\sin(k\,\pi\,v) & \cos(k\,\pi\,v)-\cos(l\,\pi\,v) & -\sin(l\,\pi\,v) & 0 \\ 0 & \sin(l\,\pi\,v) & 1-\cos(l\,\pi\,v) & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Pomocou balíka procedúr LinearAlgebra môžeme tiež vykonávať operácie s maticami, vektormi a skalármi.

```
> with(LinearAlgebra):
Ax := <1.00004,1.99987,-0.00012>:
b := <1.,2.,0.>:
Add(Ax,b,1,-1);
```

$$\begin{bmatrix} 0.0000400000000000400036 \\ \text{-}0.0001299999999999963480 \\ \text{-}0.000120000000000000004 \end{bmatrix}$$

Ďalším príkazom vykonávajúcim súčin matíc je príkaz Multiply

```
> s := <3|-2|7>;
```

$$s := [3, \text{-}2, 7]$$

```
> b := <x,y,z>;
```

$$b := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$

```
> d:= Multiply(s,b);
```

$$d := 3\,x - 2\,y + 7\,z$$

Výpočet inverznej matice

Inverzná matica je formálne označená ako matica umocnená na -1. Správnosť výpočtu si môžeme overiť súčinom matice a jej inverznej matice, ktorého výsledkom je jednotková matica E.

```
> K:=Matrix([[1,1,0,0],[0,0,1,1],[1,1,1,0],[0,1,0,1]]);
```

$$K := \begin{bmatrix} 1 & 1 & 0 & 0 \\ 0 & 0 & 1 & 1 \\ 1 & 1 & 1 & 0 \\ 0 & 1 & 0 & 1 \end{bmatrix}$$

```
> L:=K^(-1);
```

$$L := \begin{bmatrix} 2 & 1 & -1 & -1 \\ -1 & -1 & 1 & 1 \\ -1 & 0 & 1 & 0 \\ 1 & 1 & -1 & 0 \end{bmatrix}$$

```
> E:=K.L;
```

$$E := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Výpočet inverznej matice sa dá realizovať aj pomocou príkazu Inverse vo zvyškovej triede modulo *n*, alebo pomocou príkazu balíka LinearAlgebra - MatrixInverse

```
> X := Matrix([[1,2,3],[1,3,0],[1,4,3]]);
```

$$X := \begin{bmatrix} 1 & 2 & 3 \\ 1 & 3 & 0 \\ 1 & 4 & 3 \end{bmatrix}$$

```
> Y := Inverse(X) mod 5;
```

$$Y := \begin{bmatrix} 4 & 1 & 1 \\ 2 & 0 & 3 \\ 1 & 3 & 1 \end{bmatrix}$$

```
> Z := X.Y;
```

$$Z := \begin{bmatrix} 11 & 10 & 10 \\ 10 & 1 & 10 \\ 15 & 10 & 16 \end{bmatrix}$$

```
> with(LinearAlgebra):
MatrixInverse(<<a,b>|<c,d>>);
```

$$\begin{bmatrix} \frac{d}{a\,d-c\,b} & -\frac{c}{a\,d-c\,b} \\ -\frac{b}{a\,d-c\,b} & \frac{a}{a\,d-c\,b} \end{bmatrix}$$

Výsledok je niekedy potrebné zjednodušiť.

```
> with(LinearAlgebra):
R := Matrix([[cos(alpha),-sin(alpha)],[sin(alpha),cos(alpha)]]);
```

$$R := \begin{bmatrix} \cos(\alpha) & -\sin(\alpha) \\ \sin(\alpha) & \cos(\alpha) \end{bmatrix}$$

```
> MatrixMatrixMultiply(R, Transpose(R));
```

$$\begin{bmatrix} \cos(\alpha)^2 + \sin(\alpha)^2 & 0 \\ 0 & \cos(\alpha)^2 + \sin(\alpha)^2 \end{bmatrix}$$

```
> Map(simplify,%);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Transponovaná matica

Príkazom Transpose môžeme nájsť k danej matici maticu transponovanú. Na ukážku výpočtu transponovanej matice použijeme vyššie definovanú maticu V.

```
> Transpose(V);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \\ 10 & 11 & 12 \end{bmatrix}$$

V balíku príkazov LinearAlgebra postupujeme podobne. Vytvoríme najprv maticu M, ktorej prvky sú náhodne generované čísla, a potom nájdeme maticu ku nej transponovanú pomocou príkazov Create, Transpose

```
> with(LinearAlgebra:-Modular):
M := Create(30,3,3,random,integer);
```

$$M := \begin{bmatrix} 27 & 29 & 6 \\ 18 & 6 & 19 \\ 3 & 1 & 24 \end{bmatrix}$$

```
> Transpose(30,M,inplace): M;
```

$$\begin{bmatrix} 27 & 18 & 3 \\ 29 & 6 & 1 \\ 6 & 19 & 24 \end{bmatrix}$$

Výpočet determinantu štvorcovej matice

Determinant matice je číslo, ktoré vypočítame pomocou postupného rozkladu matice až na subdeterminanty rádu 2x2, a ich následným vyčíslením. Systém Maple vykoná túto pomerne zdĺhavú a numericky náročnú operáciu použitím jednoduchého príkazu Determinant. Definujme najprv napríklad dolnú triangulárnu maticu M a potom vypočítajme jej determinant. V ďalšom príklade vypočítame determinant definovanej matice N. Maticu môžeme definovať aj priamo v príkaze na výpočet determinantu.

```
> with(LinearAlgebra):
M := Matrix(3,[[a],[b,c],[d,e,f]],shape=triangular[lower]);
```

$$M := \begin{bmatrix} a & 0 & 0 \\ b & c & 0 \\ d & e & f \end{bmatrix}$$

```
> Determinant(M);
```

$$a\,c\,f$$

```
> N:=Matrix([[3,2,1,0],[7,6,5,4],[1,0,9,8],[5,4,3,2]]);
> Determinant(N);
```

$$N := \begin{bmatrix} 3 & 2 & 1 & 0 \\ 7 & 6 & 5 & 4 \\ 1 & 0 & 9 & 8 \\ 5 & 4 & 3 & 2 \end{bmatrix}$$

$$0$$

```
> Determinant(Matrix([[3,2,1,0],[7,6,5,4],[1,0,9,8],[5,4,3,2]]));
```

$$0$$